A Spreadsheet for Determining Present Value Breakeven Point for Mortgage Alternatives

James M. Grayson, Ph.D.
Augusta State University

Peter M. Basciano, Ph.D.
Augusta State University

Abstract

For most individuals the purchase of a home is one of the largest financial decisions they will encounter during their lifetime. Typically a home purchase also involves decisions regarding mortgage financing alternatives, and often financial counselors are asked to assist in the identification of the "optimal" mortgage product. Complicating this analysis, many of the online calculators lead to different and oftentimes incorrect results. This paper presents an EXCEL spreadsheet to assist the financial counselor in advising a client on the optimal mortgage alternative using a present value breakeven point criterion.

Introduction

Cash management is one of the major financial planning areas that financial counselors are frequently engaged in to provide advice to their clients. Involved in the scope of cash management are budgeting and debt management. One of the most significant decisions under debt management is the identification of the "optimal" mortgage financing strategy for an individual. The significance of this decision stems from the long-term contractual commitment of a significant portion of an individual's budget to service the mortgage debt. In addition, this allocation of funds may impact virtually all of the other major planning areas.

In attempting to identify the "optimal" mortgage product, the individual or their financial counselor, would need to evaluate a set of mortgage financing alternatives. Within this set are various combinations of mortgage terms, fixed and variable interest rates, and differing levels of

discount points. The focus of this paper is to advise financial counselors on the development of a spreadsheet to assist in evaluating various mortgage instruments involving differing combinations of contracted interest rates and discount points. Although many online calculators are available to assist an individual in evaluating "points versus interest" decisions, the results of these calculators are inconsistent and frequently fail to take under consideration several of the key variables.

For example, one of the primary considerations in evaluating mortgage alternatives involving the payment of discount points is the tax treatment of the points (Basciano & Grayson, 2004). Discount points are treated in one of two ways by the IRS for tax deductibility. The first method of recognition allows an individual to fully deduct the cost of the discount points in the year that they are incurred. This is scenario is referred to as immediately deductible discount points. However, this tax treatment of discount points only applies in the case of a purchase of a primary residence where the mortgage is secured by the primary residence. Further guidance on the criterion for immediate deductibility is provided in IRC 461(g)(2) and revenue procedure 94-27. In the case of mortgage refinancing or mortgages related to the purchase of property other than a primary residence, the IRS requires a ratable recognition of the discount points. For practical purposes, any points paid in association with a mortgage that do not meet the criterion established in IRC 461(g)(2) and revenue procedure 94-27 require ratable recognition. For a more detailed discussion of the tax treatment associated with discount points see Curatola, Hume, and Hurst (1988) or Knight and Knight (1995). The tax treatment of the discount points is a significant variable in determining the optimal mortgage product yet the vast majority of the available calculators do not include this variable in the analysis.

Other significant, but frequently overlooked variables in the "points versus interest" decision include additional financing requirements associated with the payment of discount points and the individual's marginal tax rate (Basciano & Grayson, 2005). The purpose of this paper is to present a spreadsheet model that financial counselors can use to advise clients on the optimal mortgage product using a present value breakeven criterion. The model presented allows the counselor to compare two primary scenarios: (1) the case where the discount points are immediately deductible and (2) the case of ratable recognition. Additionally, the model allow for the consideration of two secondary scenarios: (1) the case where the payment of discount points does not result in an additional financing requirement and (2) the case where the points results in a higher financing requirement for the individual.

Methodology

One proposed method for selecting between various mortgage alternatives is to compute the present value of each alternative for a specified holding period (Anderson, Barber & Chang, 1994); (Basciano & Grayson, 2005). The decision rule for selecting the optimal mortgage product for a given individual would involve identifying the mortgage alternative with the lowest present value given their expected holding period. By definition the present value of a mortgage without discount points is always equal to the amount borrowed regardless of the individual's holding period. For example, assume a hypothetical client borrows $100,000. The present value of this mortgage at any point in time, from origination to maturity, is always equal to the $100,000.

In the case where an individual pays discount points, the present value of the mortgage would depend on the holding period. In other words, the present value would vary for each month of the mortgage. At any given point in time, the better mortgage alternative is the one with lowest calculated present value. As a result, if a financial advisor or client can specify the mortgage holding period at the time of the mortgage origination, they can determine the optimal product by identifying the mortgage alternative with the lowest present value at that particular point in time.

For practical purposes, a financial counselor or their client would calculate the breakeven point associated with the mortgage alternative and then select the appropriate product. The breakeven point corresponds to the crossover point in the present values of the mortgage alternatives. For example, in a particular case a crossover point of 55 months is calculated for the mortgage with points. This result would indicate that if the individual's expected holding period exceeded 55 months they would be better off selecting the mortgage with discount points; otherwise, they should select the mortgage without points.

As mentioned in the introduction, the proper calculation of the present value of a mortgage with points is dependent on whether the points are immediately or ratably deductible for tax purposes. In the immediate deductibility case, the present value of the mortgage for any specified holding period (H) is computed as a function of (1) the after-tax cost of the discount points at time zero (the time of the mortgage origination), plus (2) present value of the after-tax payments over the specified holding period (time 0 to H), plus (3) the present value of the loan outstanding at the end of the specified holding period (H). Mathematically, the present value of the mortgage at time (H) equals:

$$PV = DP\ (1 - T)B_0 + \sum_{t=1}^{H} \left(\frac{P_t - I_t(1 - T)}{(1 + k)^t} + \frac{B_H}{(1 + k)^t}\right)$$

(Eq. 1)

where:

PV = present value of the loan with discount points
DP = discount points paid as a percentage of the loan amount
T = marginal tax rate of the borrower
B_0 = original loan balance
P_t = monthly principal payment at time t
I_t = monthly interest payment at time t
B_H = the principal balance outstanding on the loan at time H
k = after-tax interest rate on loan without discount points of identical maturity

Equation 1 requires modification in the event that the points require ratable recognition. In the case of ratably deductible discount points, the present value of the mortgage at any specified holding period (H) is a function of (1) the present value of the discount points, plus (2) the present value of the accumulated tax shield resulting from the discount points, plus (3) present value of the after-tax payments over the specified holding period (time 0 to H), plus (4) the present value of the loan outstanding at the end of the specified holding period (H). Mathematically, the present value of the mortgage with ratably recognized discount points at time (H) equals:

$$PV = DPB_0 - \sum_{t=1}^{H} \left(\frac{\left(\frac{DPB_0}{n}\right) * T}{(1+k)^t}\right) + \sum_{t=1}^{H} \left(\frac{P_t - I_t(1-T)}{(1+k)^t}\right) + \frac{B_H}{(1+k)^t}$$

(Eq. 2)

It is a common occurrence for an individual to roll part, or all, of their mortgage origination and closing costs into their loan and the payment of discount points in many cases results in an additional financing requirement. In practice, the other origination fees and closing costs are more likely "identified" as the "rolled" in portion rather than the discount points. However, the payment of discount costs still resulted in a greater financing requirement whether or not we chose to acknowledge them as financed or not and therefore is appropriately considered an opportunity cost. Consequently, to incorporate the opportunity cost of the points the initial loan balance is increased by the cost of the points, or relevant portion thereof, in the event that they resulted in an additional funding requirement.

It is also important to note that breakeven points calculated in equations 1 and 2 assume that an individual is not subject to private mortgage insurance (PMI). Typically no PMI is incurred as long as the initial loan to

value ratio is less than eighty-percent. In the case where discount points result in an additional financing requirement, an assumption is made that this would not result in additional PMI related expenses. If this assumption is violated the computed breakeven points are understated.

The above equations provide a mechanism to quantify the optimal mortgage product for an individual given a specified mortgage holding period. The optimal mortgage product is the alternative with the lowest calculated present value for the specified mortgage holding period (H). The problem with utilizing this approach in practice is that oftentimes an individual cannot specify the intended mortgage holding period with any degree of certainty at the time of the mortgage origination. As a result, it is frequently better to calculate the present value breakeven point of the mortgage with points as compared to the one without points. This breakeven point is then compared to the expected holding period to select the appropriate mortgage alternative. The remainder of this paper will provide instructions on designing a spreadsheet that will assist financial counselors and their clients in applying equations 1 and 2 and determining the present value breakeven points.

Spreadsheet Development

The present value of the mortgage with points is calculated at each month over the mortgage term (e.g. 360 months). The individual's breakeven point in present value terms is identified by comparing the present value of the mortgage with points in each month to the present value of the mortgage without points. Once a breakeven point occurs between the present values, the mortgage with points would continue as the better option until the maturity of the loan. The mortgage with points is optimal in the event that an individual intends to hold the mortgage beyond the breakeven. Conversely, the mortgage without points is the better alternative if the individual intends to hold the mortgage less than the breakeven point.

The spreadsheet model shown is developed in three sections. The first section (Table 1) contains the model assumptions and variables that are entered for a specific mortgage scenario. This information is grouped in five panels as illustrated in Table 1. Additionally, there are three types of worksheet cells in this section. One set of cells is specific to the borrower and the mortgage alternative, e.g. the marginal tax rate, desired loan amount, loan term, and so forth. The second set, shaded in gray, is derived from the borrower and mortgage information and are used to compute the present value breakeven points. An example of this type of variable is the after tax mortgage rate without points which is a function of the individual's tax rate and the interest rate on the mortgage without points. Also shown in top section of the spreadsheet, shaded in lighter gray, are the outcome cells reporting the present value breakeven point for each tax scenario based on the prior inputs.

Table 1
Model Assumptions and Variables

	A	B	C	D	E	F	G	H	I	J	K	L
1												
2	Client Information:				Immediately Deductible Points:				Ratably Deductible Points			
3	Marginal Tax Rate	Desired Loan Amount	Desired Mortgage Term		Cost of Points	After Tax Cost of Points	Breakeven Point in Months		Cost of Points	Monthly Recognition of Points	Monthly Tax Shield	Breakeven Point in Months
4	33%	$100,000.00	360		$1,000	$670	55		$1,000	$2.78	$0.92	85
5												
6	Mortgage Without Points Information:					Mortgage With Points Information:						
7	Interest Rate	Monthly Payment	After Tax Monthly Payment	After Tax Interest Rate		Adjusted Amount Financed	Interest Rate	Points Paid	Cost of Point	Monthly Payment	After Tax Monthly Payment	After Tax Interest Rate
8	6.00%	$599.55	$401.70	4.02%		$100,000	5.750%	1	$1,000	$583.57	$390.99	3.85%

Table 2 shows the formulas for the cells in Table 1.

Table 2
Data and Formulae for Model Assumptions and Variables

Cell	Purpose	Formula
	Client Information	
Cells A4:C4 are User defined		
	Immediately Deductible Points	
E4	Cost of Points	=I8
F4	After Tax Cost of Points	=E4*(1-A4)
G4	Breakeven Point in Months	=MATCH(B4,K5:K374,-1)+1
	Ratably Deductible Points	
I4	Cost of Points	=I8
J4	Monthly Recognition of Points	=I8/360
K4	Monthly Tax Shield	=J4*A4
L5	Breakeven Point in Months	=MATCH(B4,L379:L738,-1)+1
	Mortgage Without Points Information	
A8	Interest Rate	User defined
B8	Monthly Payment	=PMT(A8/12,C4,-B4)
C8	After Tax Monthly Payment	=B8*(1-A4)
D8	After Tax Interest Rate	=A8*(1-A4)
	Mortgage With Points Information	
F8	Adjusted Amount Financed	User defined
G8	Interest Rate	User defined
H8	Points Paid	User defined
I8	Cost of Point	=F8*H8/100
J8	Monthly Payment	=PMT(G8/12,C4,-F8)
K8	After Tax Monthly Payment	=J8*(1-A4)
L8	After Tax Interest Rate	=G8*(1-A4)

The second section of the model shows the cells used to compute the present value at each period for the scenario of immediate deductible discount points. The organization of the spreadsheet is shown in Table 3 and the underlying formulas are shown in Table 4.

Table 3
Immediately Deductible Discount Point Model

	A	B	C	D	E	F	G	H	I	J	K
11											
12	Immediately Deductible Discount Points										
13											
14	Period	Balance	Payment	Principal	Interest	After Tax Interest	Loan Balance	Present Value of Points	Present Value of Payments	Present Value of Loan Balance	Total Present Value
15	1	$100,000.00	$583.57	$104.41	$479.17	$321.04	$99,895.59	$670.00	$ 424.03	$ 99,562.06	$100,656.09
16	2	$99,895.59	$583.57	$104.91	$478.67	$320.71	$99,790.69	$670.00	$ 846.80	$ 99,125.43	$100,642.24
17	3	$99,790.69	$583.57	$105.41	$478.16	$320.37	$99,685.28	$670.00	$ 1,268.33	$ 98,690.12	$100,628.45
18	4	$99,685.28	$583.57	$105.91	$477.66	$320.03	$99,579.36	$670.00	$ 1,688.62	$ 98,256.10	$100,614.72

Table 4
Data and Formulae for Immediate Deductible Discount Point Model

Cell	Formula	Copied to:
A15	=1	—
A16	=A15+1	A17:A374
B15	=F8	—
B16	=G15	B17:B374
C15	=PMT($G$8/12,$C$4,-$B$4)	C16:C374
D15	=C15-E15	D16:D374
E15	=B15*$G$8/12	E16:E374
F15	=E15*(1-$A$4)	F16:F374
G15	=B15-D15	G16:G374
H15	=$F$4	H16:H374
I15	=(D15+F15)/(1+($D$8/12))^A15	—
I16	=(D16+F16)/(1+($D$8/12))^A16+I15	I17:I374
J15	=G15/(1+($D$8/12))^A15	J16:J374
K15	=SUM(H15:J15)	K16:K374

The third section of the model shows the cells used to compute the present value at each period for the scenario of ratably deductible discount points. The organization of the spreadsheet is shown in Table 5 and the underlying formulas are shown in Table 6.

Table 5
Ratably Deductible Discount Point Model



	A	B	C	D	E	F	G	H	I	J	K	L
375												
376		Ratably Deductible Discount Points										
377												
378	Period	Balance	Payment	Principal	Interest	After Tax Interest	Present Value of Tax Shield	Loan Balance	Present Value of Future	Present Value of Payments	Present Value of Loan Balance	Total Present Value
379	1	$100,000.00	$883.57	$384.41	$470.17	$321.84	$6.88	$99,615.59	$990.13	$434.03	$99,562.96	$100,885.21
380	2	$99,615.59	$883.57	$384.91	$470.67	$320.75	$13.73	$99,230.68	$980.37	$846.80	$99,125.43	$100,979.51
381	3	$99,230.68	$883.57	$385.41	$470.16	$320.37	$13.54	$98,845.26	$967.46	$1,268.83	$98,690.12	$100,852.00

Table 6
Data and Formulae for Ratable Deductible Discount Point Model

Cell	Formula	Copied to:
A379	=1	—
A380	A379+1	A381:A738
B379	=F8	—
B380	=H379	B381:B738
C379	=PMT($G$8/12,$C$4,-$B$4)	C380:C738
D379	=C379-E379	D380:D738
E379	=B379*$G$8/12	E380:E738
F379	=E379*(1-$A$4)	F380:F738
G379	=$K$4/(1+$D$8)^A379	—
G380	=G379+$K$4/(1+$D$8)^A380	G381:G738
H379	=B379-D379	H380:H738
I379	=$I$8-G379	I380:I738
J379	=(D379+F379)/(1+($D$8/12))^A379	—
J380	=(D380+F380)/(1+($D$8/12))^A380+J379	J381:J738
K379	=H379/(1+($D$8/12))^A379	K380:K738
L379	=SUM(I379:K379)	L380:L738

The outcome variables for this spreadsheet model are the breakeven points shown in Table 1 with the underlying formula shown in Table 2. The logic behind the formula is to identify the time period at which the total present value becomes equal to the loan amount without points. For example, the scenario presented in Table 1 evaluates two potential financing arrangements for an individual subject to a 33% marginal tax rate: (1) $100,000 loan at 6% interest for 360 months with no discount points and (2) $100,000 loan at 5.75% interest with one discount point. The result cells of the spreadsheet presented in Table 1 and highlighted in light gray, report the breakeven points for both the immediate and ratably deductible scenarios.

Extensions

The spreadsheet presented in Table 1 is flexible enough to evaluate the opportunity costs associated with the payment of discount points. In the event that the payment of the discount points results in an additional borrowing requirement, for example financing of other closing costs, the loan amount for the mortgage with points is adjusted (cell F8). For example, assume that the hypothetical borrower in the earlier example requires an additional $1,000 in financing if they select the mortgage alternative with points. In cell B4 they would enter $100,000 corresponding to the loan amount without points and in F8 they would enter $101,000 corresponding to

the total financed amount including the "rolled" in closing costs. Table 7 below indicates the results associated with this modification in the hypothetical individual's borrowing requirement. In addition to evaluating the impact of the additional financing requirement associated with discount points, an individual could use the spreadsheet to evaluate the sensitivity of the present value breakeven point to the other decision variables.

Table 7
Extension to Evaluate Additional Financing Requirements and Discount Points

	A	B	C	D	E	F	G	H	I	J	K	L
1												
2	Client Information:				Immediately Deductible Points:				Ratably Deductible Points			
3	Marginal Tax Rate	Desired Loan Amount	Desired Mortgage Term		Cost of Points	After Tax Cost of Points	Breakeven Point in Months		Cost of Points	Monthly Recognition of Points	Monthly Tax Shield	Breakeven Point in Months
4	33%	$100,000.00	360		$1,010	$677	176		$1,000	$2.81	$0.93	245
5												
6	Mortgage Without Points Information:				Mortgage With Points Information:							
7	Interest Rate	Monthly Payment	After Tax Monthly Payment	After Tax Interest Rate	Adjusted Amount Financed	Interest Rate	Points Paid	Cost of Point	Monthly Payment	After Tax Monthly Payment	After Tax Interest Rate	
8	6.00%	$599.55	$401.70	4.02%	$101,000	5.750%	1	$1,010	$589.41	$394.90	3.85%	

Conclusion

This paper presents a spreadsheet model that allows financial counselors to help their clients evaluate the trade-offs between mortgage financing alternatives including the option of paying discount points. More particularly, the spreadsheet provides the calculation of the present value breakeven points for both immediately and ratably deductible discount points. In addition, the flexibility of the spreadsheet also facilitates an analysis of the impact of additional borrowing requirements associated with the payment of discount points on the individual's breakeven point. The selection of the "optimal" mortgage product is extremely important given that the mortgage will result in an allocation of a significant portion of the individual's budget over a long time horizon. Consequently, it is important to have available tools to assist in the accurate calculation of the potential benefits or costs associated with the various mortgage alternatives. Unfortunately, many of the available tools fail to consider several of the significant variables and provide suspect results. For example, many of the online calculators analyzing the "points versus interest" decision fail to take into consideration the tax treatment of discount points, the tax shields associated with interest payments and discount points, and the potential additional financing requirements resulting from the payment of discount points. The spreadsheet presented in this paper is intended to address these concerns and to assist financial counselors and their clients in developing more accurate estimates of the present value breakeven points.

References

Anderson, G. A., Barber, J. R., & Chang, C. (1994). A decision model for fixed-rate mortgages. *Real Estate Finance*, *10(4),* 69-74.

Basciano, P., & Grayson, J. (2004, October 7-8). Revisiting the points versus interest rate decision for residential mortgages. *Proceedings of the 2004 Southeast Institute for Operations Research and the Management Sciences (SE INFORMS),* Myrtle Beach, SC.

Basciano, P., & Grayson, J. (2005) Beware of additional financing requirements resulting from mortgage discount points. Unpublished working paper.

Curatola, A. C., Hume, E. C., & Hurst, B. (1988, March). Points on home mortgage loans. *The Tax Advisor*, 202-217.

IRC Section 461(g)(2) and Revenue Procedure 94-27, 1994-1 CB 613.

IRC Section 461(g)(1).

Knight L. G., & Knight, R. A. (1995, February). Deductibility of mortgage points and interest. *Journal of Accountancy*, 75-78.

Contact Information: James M. Grayson, Ph.D., Associate Professor of Management, Augusta State University, College of Business Administration, 2500 Walton Way, Augusta, GA 30904-2200; Phone:(706) 667-4532; Email: jgrayson@aug.edu

Peter M. Basciano, Ph.D., Assistant Professor of Finance, Director of CFP® Certification Education Program, Augusta State University, College of Business Administration, 2500 Walton Way, Augusta, GA 30904-2200; Phone: (706) 667-4535; Email: pbasciano@aug.edu